Lectures 2-3

Functional languages

Iztok Savnik, FAMNIT

March, 2021.

Literature

John Mitchell, Concepts in Programming Languages, Cambridge Univ Press, 2003 (Chapter 7)

Michael L. Scott, Programming Language Pragmatics (3rd ed.), Elsevier, 2009 (Chapters 10)

Outline

- History
- Values
- Lambda abstraction
- Functions
- Bindings, scope and lifetime
- Recursion
- Polymorphism
- Higher-order functions
- Type annotations
- Pattern-matching

Models of computation

- In 1930s many mathematicans were dealing with the formal notion of algorithm
  - Turing, Church, Kleene, Post, ...
  - Formalizations of »effective procedure«
- Functional models:
  - Lambda calculus, General recursive functions. Combinatory logic. Abstract rewriting systems
- Sequential models:
  - Finite state machines. Pushdown automata. Random access machines. Turing machines
- Concurrent models:
  - Cellular automaton. Kahn process networks. Petri nets. Synchronous Data Flow. Interaction nets. Actor model

Models of computation

- Church's thesis says that all such formalizations are equally powerful
- Foundations of imperative and functional languages
  - Turing machine
  - Lambda calculus

Functional languages

- In recent years functional languages have become increasingly more popular
  - Scientific as well as bussiness applications
- Functional languages have a great deal in common with imperative and object-oriented relatives
  - Names, scoping, expressons, types, recursion, ...
- We will learn common concepts of PL in different paradigms
  - Functions, parameter passing, blocks, ...
- Specific concepts of particular computation models will also be presented
  - Iteration, recursion, OO paradigm, ...

Values

- Atomic types
  - integer, real, boolean, string
  - Simple value is an instance of atomic type
- Functional languages use values
- Imperative languages use variables
- Ocaml includes atomic types:
  - Integer, float, char, string
  - Operations are bound to particular types
  - Derivation of expression type is easier

```
# 7;;
- : int = 7
# 5.3;;
- : float = 5.3
# 'c';;
- : char = 'c'
# "spring";;
- : string = "spring"
```

Lists

- Values of some type can be stored into lists
  - List is either empty [], or, it includes valus of fixed type
- List has a head and a tail:
  - 1 is head and [2; 3] a tail
  - [1; 2; 3] ≡ 1::[2; 3] ≡ 1::2::3::[]
  - Operator :: corresponds to Lisp cons operator
- Two lists can be concatenated using operator append @
  - [1]@[2; 3] ≡ [1; 2; 3]
- Implementation details & other aspects
  - Lectures on (1) Imperative lang. and (2) Types

```
# [ 1 ; 2 ; 3 ] ;;
- : int list = [1; 2; 3]
```



Products

- Products are defined as in mathematics
  - Interpretation of a product type is the Cartesian product of the interpretations of types that comprise product
  - Instances of products are pairs, triples, n-tuples

```
# ( 65 , ’B’ , "ascii") ;;
- : int * char * string = 65, ’B’, "ascii"
```

```
# ( 12 , "October") ;;
- : int * string = 12, "October"
```

```
# fst;;
- : ’a * ’b -> ’a = <fun>
# fst ( "October", 12 ) ;;
- : string = "October"
```

`let` statement

`let x=N in M` $\equiv$ $(\lambda x.M)\ N$

- λ-abstraction $\lambda x.M$ with a given argument N
- Used for binding the name and the value (later)

- Forms of `let` in Ocaml
  - Global [simultanous] declaration
  - Local [simultanous] declaration

```
let name₁= expr₁
...
and nameₙ= exprₙ
in expr
```

```
let name₁= expr₁
...
and nameₙ= exprₙ
```

`let` statement

- Examples

```
# let a = 3.0 and b = 4.0;;
val a : float = 3.
val b : float = 4.
# a;;
- : float = 3.
```

```
# let a = 3.0 and b = 4.0 in sqrt (a*.a +. b*.b);;
- : float = 5.
# a;;
Error: Unbound value a
```

```
# let c = (let a = 3.0 and b = 4.0 in sqrt (a*.a +. b*.b));;
val c : float = 5.
```

Functions

- Function is lambda abstraction $\lambda x.M$
  - x is a parameter and M is the body of a function
- Function expression is composed of a parameter x and a function body M
  - Function parameter is formal

```
function x -> M
```

- Example:
  - $\lambda x.x*x$
  - $(\lambda x.x*x)\ 5$

```
# function x -> x*x ;;
- : int -> int = <fun>
# (function x -> x * x) 5 ;;
- : int = 25
```

Functions

- Function body can be another function

```
# function x -> (function y -> 3*x + y) ;;
- : int -> int -> int = <fun>
```

- The result of a function can again be a function

```
# (function x -> function y -> 3*x + y) 5 ;;
- : int -> int = <fun>
```

```
# (function x -> function y -> 3*x + y) 4 5 ;;
- : int = 17
```

Function

- The arity of function is number of its parameters
  - Functions have single parameters in OCaml
  - This is called Curry form of function

$f\ (g, x) = g(x)$

$f_{curry} = \lambda g.\lambda x.g\ x$

- Single parameter can also be a tuple or a record
  - Curry form of the same function

```
# function (x,y) -> 3*x + y ;;
- : int * int -> int = <fun>
```

```
# function x -> function y -> 3*x + y;;
- : int -> int -> int = <fun>
# (function x -> function y -> 3*x + y) 4 5;;
- : int = 17
```

Function values

- Function is treated as a value
  - Using `let` statement to define binding between function and name
- Examples:

```
# let succ = function x -> x + 1 ;;
val succ : int -> int = <fun>
# succ 420 ;;
- : int = 421
# let g = function x -> function y -> 2*x + 3*y ;;
val g : int -> int -> int = <fun>
# g 1 2;;
- : int = 8
```

Function values

- Alternative syntax for function definition in Ocaml

```
let name p_1 p_2 ... = <function-body>
let name = function p_1 -> ... -> function p_n -> <function-body>
```

- Example:
  - h1(y) = 2 + 3*y
  - h2(x) = 2*x + 6

```
# let s x = x + 1;;
val s : int -> int = <fun>
# let g x y = 2*x + 3*y ;;
val g : int -> int -> int = <fun>
# let h1 = g 1 ;;
val h1 : int -> int = <fun>
# let h2 = function x -> g x 2 ;;
val h2 : int -> int = <fun>
# h2 2 ;;
- : int = 10
```

Bindings

- A binding is an association between two things, such as a name and the thing it names
- let statement binds value with its name
  - let x = M in N ≡ $(\lambda x.N)M$
- Value can be defined globally or locally
  - Global definition is accessible in global context
  - Local definition is seen solely in local context

```
# let x = 3 ;;
- : int = 3
```

```
# let x = 3 in x * x ;;
- : int = 9

# (let x = 3 in x * x) + 1 ;;
- : int = 10
```

Blocks

```
{ int x = 2;
   { int y = 3;
     x = y+2;
   }
}
```

- Most modern programming languages provide some kind of blocks
  - Block is program region that includes begin and end
  - Blocks have local variables
  - Global variable of some block is defined in some encompassing block

```
{ int x = . . . ;
  { int y = . . . ;
    { int x = . . . ;
      ...
    };
  };
};
```

- Block are used in C, C++, Java, ML, ...
- Local definition of value can hide the definition of global value

```
# let a = 1.0;;
- : float = 1
# let a = 3.0 and b = 4.0 in sqrt (a*.a +. b*.b) ;;
- : float = 5
```

Scope and lifetime

- Scope of value (or variable) is program area where it is defined
  - Value defined in some global block is defined in all subsuming blocks
- Lifetime of value (or variable) is duration of definition of value
- In many cases scope implies lifetime
  - Global declarations in C, C++ can appear locally

Static and dynamic scope

- Let some value be defined outside current block
  - Then variable is global relatively to local block
- Static scope
  - Variables are first searched in local block and then in structurally enclosing blocks
- Dynamic scope
  - Variables are searched by following function calls i.e. blocks where function was invoked
- Static: C, Schema, ML, Pascal

  Dynamic: older Lisp, macros in C

```
# let x=1;;
val x : int = 1
# let g z = x+z;;
val g : int -> int = <fun>
# let f y = let x = y+1 in g (y*x);;
val f : int -> int = <fun>
# f(3);;
- : int = 13 (or 16)?
```

Recursion

- Definition of a symbol includes reference to itself
- Recursion was first introduced in Lisp
  - McCarthy advocated to use recursion in Algol
- Lambda abstractions do not have name
  - McCarthy suggestion for naming functions in Lisp

```
(label f (lambda (x) (cond ((eq x 0) 0) (true (+ x (f (- x 1)))))))
```

  - Later they simply declared function using `define`

```
(define f (lambda (x) (cond ((eq x 0) 0) (true (+ x (f (- x 1)))))))
```

Linear recursion

- Recursion that evolves in one direction
  - Recursive call is at the end of function body
- Example: factorial

```
# let rec factorial n -> if n=1 then 1 else n * factorial (n-1);;
val factorial : int -> int = <fun>
```

- Tail recursion
  - Can be converted into iteration

```
fact 6 = 6 * fact 5
       = 6 * ( 5 * fact 4)
       = 6 * ( 5 * (4 * fact 3))
       = 6 * ( 5 * (4 * ( 3 * fact 2)))
       = 6 * ( 5 * (4 * ( 3 * ( 2 * fact 1))))
       = 6 * ( 5 * (4 * ( 3 * ( 2 * 1))))
       = 720
```

Examples

- Function `sigma` computes sum from `1` to n.
  - Linear recursion

```
# let rec sigma x = if x = 0 then 0 else x + sigma (x-1) ;;
val sigma : int -> int = <fun>
# sigma 10 ;;
- : int = 55
```

- Function `even` and `odd` return boolean value
  - Recursion in two cycles

```
# let rec even n = (n<>1) && ((n=0) or (odd (n-1)))
  and odd n = (n<>0) && ((n=1) or (even (n-1)));;
val even : int -> bool = <fun>
val odd : int -> bool = <fun>
# even 4 ;;
- : bool = true
# odd 5 ;;
- : bool = true
```

Functions on lists

- A list is a linear data structure
  - Grows in one direction
  - Linear recursion
    - Recursion guided by the structure
  - Stopping condition is the end of a list
  - Can be converted into iteration

Functions on lists

```
# let null l = (l = []) ;;
val null : 'a list -> bool = <fun>
# let rec size l =
if null l then 0
else 1 + (size (List.tl l)) ;;
val size : 'a list -> int = <fun>
# let rec reverse l =
if l=[] then []
else (reverse (List.tl l)) @ [(List.hd l)];;
val reverse : 'a list -> 'a list = <fun>
```

Functions on lists

```
# let rec member x l =
if l=[] then false
else if x = List.hd(l) then true
else member x (List.tl l);;
val member : ’a -> ’a list -> bool = <fun>
# member 3 [2;3;1];;
- : bool = true
# let rec inter(xs, ys) =
if xs=[] then []
else let x = List.hd xs
        in if (member x ys) then x :: inter(List.tl xs, ys)
            else inter(List.tl xs, ys);;
val inter : ’a list * ’a list -> ’a list = <fun>
# inter ([1;2;3],[4;2]);;
- : int list = [2]
```

Polymorphism

- Polymorphism (Greek, »many shapes«)
- Parametric polymorphism
  - Function can have »many shapes«
  - Types of parameters are variables
- Type variables
  - Variable that stands for any type
  - 'a, 'b, 'c, ...
- A form of genericity

```
# let make_pair a b = (a,b) ;;
val make_pair : 'a -> 'b -> 'a * 'b = <fun>
# let p = make_pair "paper" 451 ;;
val p : string * int = "paper", 451
# let a = make_pair 'B' 65 ;;
val a : char * int = 'B', 65
# fst p ;;
- : string = "paper"
# fst a ;;
- : char = 'B'
```

Examples:
Polymorphic functions on lists

```
# let rec member x l =
if l=[] then false
else if x = List.hd(l) then true
else member x (List.tl l);;
val member : ’a -> ’a list -> bool = <fun>
# member 3 [2;3;1];;
- : bool = true
# let rec append l1 l2 =
if null l1 then l2
else (List.hd l1)::append (List.tl l1) l2    ;;
val append : 'a list -> 'a list -> 'a list = <fun>
# append [1;2] [3;4];;
- : int list = [1; 2; 3; 4]
```

Examples

```
# let app = function f -> function x -> f x ;;
val app : ('a -> 'b) -> 'a -> 'b = <fun>
```

- Function application
  - Any function f of type 'a->'b can be passed as parameter

```
# app odd 2;;
- : bool = false
# let id x = x ;;
val id : 'a -> 'a = <fun>
# app id 1 ;;
- : int = 1
```

- Function composition
  - Any functions f and g of type 'a->'b and 'c->'a can be passed as parameters

```
# let compose f g x = f (g x) ;;
val compose : ('a -> 'b) -> ('c -> 'a) -> 'c -> 'b = <fun>
# let add1 x = x+1 and mul5 x = x*5 in compose mul5 add1 9 ;;
- : int = 50
```

Currying

- Function with more than one argument can be represented by sequence of functions with one argument
  - This procedure is called Currying
- Example

```
# let add1 x y = x + y;;
val add1 : int -> int -> int = <fun>
# add1 3 4;;
- : int = 7
```

```
# let add2 (x,y) = x + y;;
val add2 : int*int -> int = <fun>
# add2 (3,4);;
- : int = 7
```

```
# let curry f = function x -> function y -> f (x,y);;
val curry : ('a * 'b -> 'c) -> 'a -> 'b -> 'c = <fun>
# let uncurry f = function (x,y) -> f x y;;
val uncurry : ('a -> 'b -> 'c) -> 'a * 'b -> 'c = <fun>
# uncurry add1;;
- : int * int -> int = <fun>
# curry(uncurry add1);;
- : int -> int -> int = <fun>
```

Higher-order functions

- Higher-order function either takes function as the parameter, or, returns function.
  - We have already seen many examples
- Function compose is an example of higher-order function

```
# let compose f g x = f (g x) ;;
val compose : ('a -> 'b) -> ('c -> 'a) -> 'c -> 'b = <fun>
```

  - The result is compositum of two functions stated as parameters compose
    - Result is again a function

Higher-order functions

- Function map is an example of higher-order function

```
# let rec map f l =
if null l then []
else f(List.hd l)::(map f (List.tl l));;
val map : ('a -> 'b) -> 'a list -> 'b list = <fun>
# let square x = string_of_int (x*x) ;;
val square : int -> string = <fun>
# map square [1; 2; 3; 4] ;;
- : string list = ["1"; "4"; "9"; "16"]
```

- Higher-order functions can be useful as programming tool
  - Very common use recently!

Higher-order functions

- Function for_all is an example of higher-order function
  - Universal quantification for a property expressed as boolean function f on elements of list l

```
# let rec for_all f l =
if null l then true
else (f (List.hd l)) && for_all f (List.tl l);;
val for_all : ('a -> bool) -> 'a list -> bool = <fun>
# for_all (function n -> n<>0) [-3; -2; -1; 1; 2; 3] ;;
- : bool = true
# for_all (function n -> n<>0) [-3; -2; 0; 1; 2; 3] ;;
- : bool = false
```

Higher-order functions

- Folding a list
  - fold_left f a [e_1; e_2; ... ; e_n] = f (... (f (f a e_1) e_2) ... e_n).

```
# let rec fold_left f a l =
    if null l then a
    else fold_left f ( f a (List.hd l)) (List.tl l) ;;
val fold_left : ('a -> 'b -> 'a) -> 'a -> 'b list -> 'a = <fun>
```

```
# let sum_list = fold_left (+) 0 ;;
val sum_list : int list -> int = <fun>
# sum_list [2;4;7] ;;
- : int = 13
# let concat_list = fold_left (^) "";;
val concat_list : string list -> string = <fun>
# concat_list ["Hello "; "world"; "!"] ;;
- : string = "Hello world!"
```

Higher-order functions

- Function constructs function

```
# let rec iterate n f =
if n = 0 then (function x -> x)
else compose f (iterate (n-1) f) ;;
val iterate : int -> ('a -> 'a) -> 'a -> 'a = <fun>
```

- Construction of function power

```
# let rec power i n =
let i_times = ( * ) i in
iterate n i_times 1 ;;
val power : int -> int -> int = <fun>
# power 2 8 ;;
- : int = 256
```

Pattern matching

- A special feature of languages of ML family
  - Haskell, Erlang, SML, ...
  - Close to Prolog unification
  - Very complex `case` statement
- Declarative language construct !
  - Logic based languages?
  - Not functional and not imperative?
- Allows simple access to the components of complex data structures
- Functions can be defined by cases
  - Pattern matching over an argument

Pattern matching

- Patterns
  - Structure comprised of tuples, records, unions and lists including constants (of predefined types) and variables
  - Variables are »hooks« that catch the values
  - The symbol _ is called the wildcard pattern: matches to any data (as in Prolog)
- The evaluation is parametrised by data:
  - when pattern in data is recognised, the corresponding expression is evaluated.

Syntax and usage

```
match <expression> with
| <pattern_1> -> <expression_1>
| <pattern_2> -> <expression_2>
....
| <pattern_k> -> <expression_k>
```

- Patterns in match must be of same type
- Pattern must be linear - a variable can appear just once in a pattern
- Patterns in match are tested sequentially (!)
- The expression with the first match is evaluated
- List of patterns in match must be exhaustive - every value must be matched with a pattern in the list
- First pipe (character | on the first line) is optional

Examples

- Simple match:

```
# let imply v = match v with
    | (false,false) -> true
    | (false, true) -> true
    | (true, false) -> false
    | (true, true) -> true;;
val imply : bool * bool -> bool = <fun>
```

- With variable:

```
let imply2 a b = match (a,b) with
   | (true, x) -> x
   | (false,x) -> true;;
Val imply2 : bool -> bool -> bool = <fun>
```

- With wildchard pattern:

```
let imply3 a b = match (a,b) with
   | (true,false) -> false
   | _ -> true;;
let imply3 : bool -> bool -> bool = <fun>
```

Linearity and completeness

- Every pattern must be linear:

```
let equal a b = match (a,b) with
   | (x,x) -> true
   | _ -> false;;
Error: Variable x is bound several times in this matching
# equal 1 2;;
Error: Unbound value equal
```

- Every pattern must be exhaustive:

```
# let iszero x = match x with 0 -> true;;
Warning 8: this pattern-matching is not exhaustive.
Here is an example of a value that is not matched:
1
val iszero : int -> bool = <fun>
# iszero 0;;
- : bool = true
# iszero 10;;
Exception: "Match_failure //toplevel//:3:-34".
```

Combining patterns

Pattern matching is sequential:

1. 'A' is »Consonant«
2. 'A' matched by »Vowel«

```
# let char_discriminate c = match c with
   |’a’ | ’e’ | ’i’ | ’o’ | ’u’ | ’y’
   | ’A’| ’E’ | ’I’ | ’O’ | ’U’ | ’Y’ -> "Vowel"
   | ’a’..’z’ | ’A’..’Z’ -> "Consonant"
   | ’0’..’9’ -> "Digit"
   | _ -> "Other";;
val char_discriminate : char -> string = <fun>

# val char_discriminate ’A’;;
- : string = "Vowel"
# val char_discriminate ’z’;;
- : string = "Consonant"
# val char_discriminate ’$’;;
- : string = "Other"
```

Matching on arguments

- Pattern matching is used in an essential way for defining (unary) functions by cases.
- Syntax:

```
function
| <pattern_1> -> <expression_1>
....
| <pattern_k> -> <expression_k>
```

- Indeed, the construction of function <x> -> <expression>, is a definition by pattern matching using a single pattern reduced to one variable.

```
# let f = function (x,y) -> 2*x + 3*y;;
val f : int * int -> int = <fun>
```

```
# let f (x,y) = 2*x + 3*y + 4 ;;
val f : int * int -> int = <fun>
```

```
# let rec sigma = function
0 -> 0
| x -> x + sigma (x-1) ;;
val sigma : int -> int = <fun>
# sigma 10 ;;
- : int = 55
```

Named patterns

- During pattern matching, it is sometimes useful to name part or all of the pattern. This is useful when one needs to take apart a value while still maintaining its integrity.

```
# let less_rat pr = match pr with
   | ((_,0),p2) -> p2
   | (p1,(_,0)) -> p1
   | (((n1,d1) as r1), ((n2,d2) as r2)) ->
      if (n1*d2) < (n2*d1) then r1 else r2;;
val min_rat : (int * int) * (int * int) -> int * int = <fun>
```

- As a result, the value matched by the named pattern can be returned.

Pattern guards

- Guard is a conditional expression applied immediately after the pattern is matched.
- Syntax:

```
match <expression> with
....
| <pattern_i> when <condition> -> <expression_i>
....
```

- Example:

```
# let eq_rat cr = match cr with
     ((_,0),(_,0)) -> true
   | ((_,0),_) -> false
   | (_,(_,0)) -> false
   | ((n1,1), (n2,1)) when n1 = n2 -> true
   | ((n1,d1), (n2,d2)) when ((n1 * d2) = (n2 * d1)) -> true
   | _ -> false;;
val eq_rat : (int * int) * (int * int) -> bool = <fun>
```

Examples

- Size of a list:

```
# let rec length = function
   | [] -> 0
   | _::tl -> 1 + length tl;;
val length : 'a list -> int = <fun>
# length [1;2;3;4;5];;
- : int = 5
```

- Appending of two lists:

```
# let rec append = function
    | [], l -> l
    | hd::tl, l -> hd :: append (tl,l);;
val append : 'a list * 'a list -> 'a list = <fun>
# append ([1;2;3;4], [5;6;7]);;
- : int list = [1; 2; 3; 4; 5; 6; 7]
```

Examples

- Membership test:

```
# let rec contains e lst = match lst with
    | [] -> false
    | hd::_ when e = hd -> true
    | _::tl -> contains e tl;;
val contains : 'a -> 'a list -> bool = <fun>
# contains 1 [1;2;3;4;5;6];;
- : bool = true
# contains 10 [1;2;3;4;5;6];;
- : bool = false
```

- Intersection of two lists (as sets):

```
# let rec meet l1 l2 = match l1 with
    | [] -> []
    | hd::tl when (contains hd l2) -> hd :: meet tl l2
    | _::tl -> meet tl l2;;
val meet : 'a list -> 'a list -> 'a list = <fun>
# meet [1;2;3;4;5] [3;4;5;6];;
- : int list = [3; 4; 5]
```

Examples

- Union of two lists (as sets):

```
# let rec union l1 l2 = match l1 with
    | [] -> l2
    | hd::tl when (contains hd l2) -> union tl l2
    | hd::tl -> hd :: union tl l2;;
val union : 'a list -> 'a list -> 'a list = <fun>
# union [1;2;3;4;5] [4;5;6;10];;
- : int list = [1; 2; 3; 4; 5; 6; 10]
```

Type declaration

- Type is defined from simpler types using type constructors: *, |, list, array, ...
- Type definition in Ocaml
- Example:

```
type name = typedef ;;
```

type name$_1$ = typedef$_1$
and name$_2$ = typedef$_2$
...
and name$_n$ = typedef$_n$;;

```
# type int_pair = int*int;;
type int_pair = int * int
# let v:int_pair = (1,1);;
val v : int_pair = (1, 1)
```

C:

```
typedef char byte;
typedef byte_ten bytes[10];
typedef struct {int m;} A;
typedef struct {int m;} B;
A x; B y;
x=y; /* incompatible types in assignment */
```

Parametrized types

- Type declarations can include type variables
- Type variable is a variable that can stand for arbitrary type
- Types that include variables are called parametrized types or also polymorphic types
- Parametrized type in Ocaml:

```
# type ('a,'b) pair = 'a*'b;;
type ('a, 'b) pair = 'a * 'b
# let v:(char,int) pair = ('a',1);;
val v : (char, int) pair = ('a', 1)
```

```
type 'a name = typedef ;;
type ('a1 . . . 'an ) name = typedef ;;
```

Products

- Products of types T_1*T_2*...*T_n
  - Denotation: Cartesian product of sets that correspond to types $T_1T_2...T_n$
  - $I(T_1*T_2*...*T_n) = I(T_1) \times I(T_2) \times ... \times I(T_n)$
- Examples in Ocaml:
- Operations
  - fst(), snd()
  - Pattern matching

```
# let (a,b,c) = (1,"2",'3');;
val a : int = 1
val b : string = "2"
val c : char = '3'
# let first t = match t with
      x,_,_ -> x ;;
val first : 'a * 'b * 'c -> 'a = <fun>
# first a ;;
- : int = 1
```

Product: Ocaml examples

```
# type ('a,'b,'c) triple = 'a*'b*'c;;
type ('a, 'b, 'c) triple = 'a * 'b * 'c
# let t = 1,'1',"1";;
val t : int * char * string = (1, '1', "1")
# let t:(int,char,string) triple = 1,'1',"1";;
val t : (int, char, string) triple = (1, '1', "1")
```

```
# type 'param paired_with_integer = int * 'param ;;
type 'a paired_with_integer = int * 'a
# type specific_pair = float paired_with_integer ;;
type specific_pair = float paired_with_integer
# let x:specific_pair = (3, 3.14) ;;
val x : specific_pair = 3, 3.14
```

Lists

- Lists were introduced in chapter on Functional languages
- Functional and logic languages
  - Work via recursion and higher-order functions
  - In Lisp a program is a list; can extend itself at run time
  - Built-in polymorphic functions to manipulate arbitrary lists
- Lists in Lisp
  - Two pointers: First (car) and Rest (cdr)
    - Names are historical accidents
      (contents of address|decrement register)



  - Lists are implemented in this way in Lisp
    - Also in Python, Prolog
  - Lists in Lisp are heterogeneous (of different types)

Lists

- Lists in ML-family
  - Lists in ML are homogeneous (of the same type)
  - Chains of blocks including element and a pointer to the next block
    - Also Clu (Barbara Liskov, 1974), Haskell
- Lists can also be used in imperative programs
  - Implementation as an example
- Lists work best in a language with automatic garbage collection
  - many of the standard list operations tend to generate garbage

Lists

```
/* Haskell  */
{i × i | i ∈ {1, . . . , 100} ∧ i 'mod' 2 = 1}
/* Python */
[i*i for i in range(1, 100) if i % 2 == 1]
```

- List comprehensions
  - $\{i * i \mid i \in \{1, . . . , 100\} \wedge i \bmod 2 = 1\}$
  - Haskell, Erlang, Python, and F#
- Higher-order functions in Ocaml
  - Iterators, aggregates, filters, selection, ...
  - Declarative programming?

```
val map : ('a -> 'b) -> 'a list -> 'b list
val fold_left : ('a -> 'b -> 'a) -> 'a -> 'b list -> 'a
val fold_right : ('a -> 'b -> 'b) -> 'a list -> 'b -> 'b
val for_all : ('a -> bool) -> 'a list -> bool
val exists : ('a -> bool) -> 'a list -> bool
val find : ('a -> bool) -> 'a list -> 'a
val filter : ('a -> bool) -> 'a list -> 'a list
...
```

Unions

- Type constructed by union
  - Make a new set by taking the union of existing sets
  - $I(T_1|T_2|...|T_n) = I(T_1) \cup I(T_2) \cup \ldots \cup I(T_n)$
- Unions in Ocaml
- Operations:
  - Construction of instance
  - Pattern matching

```
type name = ...
    | Name_i ...
    | Name_j of t_j ...
    | Name_k of t_k * ...* t_l ... ;;
```

```
# type coin = Heads | Tails;;
type coin = | Heads | Tails
# Tails;;
- : coin = Tails
```

Example: Tarot

```
# type suit = Spades | Hearts |
              Diamonds | Clubs ;;
# type card =
      King of suit
    | Queen of suit
    | Knight of suit
    | Knave of suit
    | Minor_card of suit * int
    | Trump of int
    | Joker ;;
```

```
# King Spades ;;
- : card = King Spades
# Minor_card(Hearts, 10) ;;
- : card = Minor_card (Hearts, 10)
# Trump 21 ;;
- : card = Trump 21
```

Example: Tarot

```
# let rec interval a b = if a = b then [b] else a :: (interval (a+1) b) ;;
val interval : int -> int -> int list = <fun>
# let all_cards s =
     let face_cards = [ Knave s; Knight s; Queen s; King s ]
     and other_cards = List.map (function n -> Minor_card(s,n)) (interval 1 10)
     in face_cards @ other_cards ;;
val all_cards : suit -> card list = <fun>
# all_cards Hearts ;;
- : card list =
[Knave Hearts; Knight Hearts; Queen Hearts; King Hearts;
 Minor_card (Hearts, 1); Minor_card (Hearts, 2); Minor_card (Hearts, 3);
 Minor_card (Hearts, ...); ...]
```

Pattern matching on unions

```
# let string_of_suit = function
        Spades   -> "spades"
      | Diamonds -> "diamonds"
      | Hearts   -> "hearts"
      | Clubs    -> "clubs";;
val string_of_suit : suit -> string = <fun>
# let string_of_card = function
        King c             -> "king of "^ (string_of_suit c)
      | Queen c            -> "queen of "^ (string_of_suit c)
      | Knave c            -> "knave of "^ (string_of_suit c)
      | Knight c           -> "knight of "^ (string_of_suit c)
      | Minor_card (c, n) -> (string_of_int n) ^ "of "^(string_of_suit c)
      | Trump n            -> (string_of_int n) ^ "of trumps"
      | Joker              -> "joker";;
val string_of_card : card -> string = <fun>
```